JACG NEWSLETTER

Vol.4 No.1

Sep. 1984

Single Copy Price $1.50

THE JERSEY ATARI COMPUTER GROUP

KEN ROSER
CARTRIDGES

REPAIRS:
DEPENDABLE PROTECH

168 MAIN ST CHATHAM

633-5888

From the Editor's Desk...

This edition of the newsletter marks the beginning of Volume 4. That means we have officially published 48 previous issues. For a personal computer user group that is a real benchmark. Things move so fast in this electronics world that four years of anything represents tremendous evolution. We thank those of you who have helped our group grow to the magnitude and sensitivity it has.

Unquestionably the key person responsible for the health enjoyed by JACG is Dick Kushner. He was one of our founding fathers, he showed the way as editor of the first newsletters, and has given us sage advice and firm direction as our president. He ends his tenure next month, earning a well-deserved respite from the almost daily demands of overseeing such a large organization. We have grown from a handful to over 550 active members. Thankfully, Dick has agreed to remain on the Executive Committee for one year as an ex officio member sharing with his successor the tricks he has acquired over the years.

Those of us who have gotten to know Dick on a personal basis will let you know he is a warm and caring person. He not only is there when you need him, I have often found him anticipating my needs before I knew I had them. Dick is, by the way, a research scientist with Bell Telephone Laboratories and holds a Ph.D. in Chemical Engineering among his other not small accomplishments. His latest effort, Basic Atari BASIC, a text written with James Coan is, in my opinion, the current definitive work on learning the workings of the Atari computer.

It would be impossible to catalog the things Dick has done for us. Such a list would be long and impressive. There is no way to repay someone for having generously dedicated so much of himself to our benefit except to say "Thanks, Dick. We really appreciate it."

Ed. note: Okay, Kushner, see you at the next Exec. COMM. meeting. Don't forget your report.

Frank Pazel

Frank Pazel
Editor-in-Chief, JACG Newsletter

In This Issue

MARK YOUR CALENDARS!!

JACG Meeting Schedule

October 13, 1984
November 10, 1984
December 8, 1984
January 12, 1985

FORTH
THU SEPT 26 7:30PM

CELLULAR AUTOMATA

or
Mutant Spawn of the Space Invaders

by Kirk McDonald - JACG

The Aliens eventually grow weary of the frustration of attacking Earth in suicide waves. Our defense is too effective, thanks to you, Jersey Atarians. But before the Aliens withdraw they conceive of a new form of assault, a kind of genetic roulette. Their mother ship emits a cloud of spores which obey a primitive genetic code as to how each new generation will arise from the previous. Due to the vigorous counter-attack from Earth the Aliens are forced to depart before their genetic engineering is complete. Even the Aliens don't know which codes produce spores that may someday grow into new breeds of super-Aliens, or those that die a lonely death in the cold of space. Or that grow uncontrollably turning the universe into a sort of Sargasso Sea inhospitable to Aliens and Earthlings alike. To provide counter-measures against these mutant spawn of the Space Invaders we must explore the significance of the Aliens' genetic codes to determine which are malignant, or benign.....

If you wish, you might just type in the listing at the end of the article and run it. Follow the prompts to enter several parameters and watch the patterns fill the screen (slowly). The Action! program discussed below runs about 300 times faster, filling the screen with 80 generations of spores in 2 seconds. It also offers a printer dump if you find an interesting specimen you wish to preserve. The 4 Figures are examples of spore growth which may be studied with these programs.

As the Action! program listing is rather lengthy it has been made available as file AUTOM.ACT on a J.A.C.G. library diskette. Check with the disk librarian to see where it is located. You must have an Action! language cartridge and an 800 or 800XL computer to run this program. The library diskette also includes a text file AUTOM.DOC which can be read by the Action! editor, and reviews the menu options of the Action! program in case they are not self-explanatory. The printer dump is written for the Epson FX-80 which has the capability of user-defined character sets which speed up the screen dump.

The programs will allow you to investigate an interesting conjecture by Stephen Wolfram of the Institute for Advanced Study at Princeton, N.J. He suggests that there are 4 general types of growth patterns of the spores, only one of which produces the super-Aliens. But even from a random intial configuration of the spores there can arise complex, self-replicating structures. See Figure 2. The implications of this phenomenon are little understood, but seem rather provocative. With the help of your Atari computer prepare to explore a new frontier of knowledge!

The remainder of this article is a somewhat technical exposition of the idea behind the genetic codes of the mutant spores, the so-called <u>cellular automata</u>. These are prescriptions of how each succeeding generation of spores is related to the previous generation. Along with others I have found surprising esthetic and mathematical appeal in these simple devices. The concepts of cellular automata are presented in the March 1984 issue of Scientific American, and in much greater detail in the January 1984 issue of Physica D (Nonlinear Phenomena) which you might find at the physics department library of a university.

We picture a generation of spores as a line of cells each of which may or may not be occupied by a spore. In the Basic version there is only one kind of spore but in the Action! program there can be up to 3 types, shown on your screen in different colors. For the case of one type of spore a generation can be represented as a binary number:

...10011101010001....

A 1 indicates the presence of a spore in a cell. The genetic code governing the growth of the spores' is based on the contents of each group of n adjacent cells, called the parents. In the Basic program n = 5, while n = 3, 5 or 7 in the Action! program. The code determines the occupant of one cell, called the daughter, in the next generation. The daughter cell is located below the center of the group of parent cells.

The case of 3 parents and only one type of spore is the simplest. There are 8 possible variations of the contents of the 3 parent cells:

111 110 101 100 011 010 001 000.

The genetic code, or <u>automaton</u>, consists of 8 digits, each corresponding to one of the 8 possible groups of parent cells. Each digit takes on a value of 0 or 1 indicating whether in the next generation the daughter cell is empty or has a new spore. That is, if the contents of the parent cells are of the ith type in the list above then refer to the ith digit of the automaton code to learn the fate of the daughter cell. This algorithm is applied to all possible groups of 3 adjacent cells when going from one generation to the next.

An automaton that I find amusing can be written

01001001

meaning that only if the parent cell group is 110, 011, or 000 will there be a spore in the next generation. Applying the automaton to the generation of spores shown in the preceding paragraph we show the next generation as a binary number on the next line, and so on for succeeding generations:

```
...1001110101001...
....00101000000....
.....000001111.....
......1110100......
.......01000.......
........001........
..........0........
```

In this example each generation has 2 fewer cells than the previous, due to "edge effects". The crafty Aliens circumvented this problem by creating a special class of immutable spores which are placed on the ends of the string of cells. Taking into account this feature the 7 generations of spores shown above would now be:

```
1001110101001
1001010000001
1000000111101
1011110101001
1010100000001
1000001111101
1011101001001
   ..etc..
```

As one masters the concept of the automaton the conclusion is soon reached that calculating the generations is an excellent job for the computer. Figure 1 shows a printer dump of 80 generations of spores calculated with the automaton discussed above. The field of spores is 160 cells wide. The program uses the GRAPHICS 6 memory as storage for the spores, which sets the field limits as stated. The computer can of course calculate an arbitrary number of generations, and a scrolling display is availble in the Action! version so that you can determine the fate of the cellular universe. In the Basic program you can continue for as many screens of 80 generations as you wish.



You can readily verify that there are
2 to the (2 to the 3rd) = 256
different possible automata for the case of 1 kind of spore and a 3 cell code. But in the general case of k kinds of spores and an n cell code there are
k+1 to the (k+1 to the nth) = lots
of different codes. An important insight was given by Stephen Wolfram. He claims that representatives of all 4 general classes of automata can be produced by a simpler rule than that given above. Namely, for n parents just add up the contents of the n cells, yielding a number bewteen 0 and n*k. For each of these sums, the automaton tells us what kind of spore if any will populate the daughter cell in the next generation. Wolfram calls such automata "totalistic" in that they are based only on the total number of spores in the parent cells, and not on the distribution of spores within those cells. The example of an automaton given in the preceding paragraph is not totalistic, although personally I find it quite interesting.

The Basic program appended to this article can calculate all 64 types of totalistic automata for 1 kind of spore with 5 parent cells. The most "interesting" automaton is given by the code

20(decimal) = 010100(binary).

The interpretation of this totalistc code is different than for the general case discussed several paragraphs above. The digits in the binary form of this code, from right to left, correspond to the sum of the spores in the parent cells being 0, 1, 2, 3, 4 or 5 and indicate what the daughter cell will contain. Hence only if the sum of the 5 cells is 2 or 4 will there be a spore in the next generation. If we apply this automaton to the initial generation of spores listed above the result would be:

```
1001110101001
1000010011101
1000001000101
1000000010101
1000000011001
1000000111001
1000001000001
1000000000001
   ..etc..
```

All subsequent generations will be the same as the last shown. Note that to allow for "edge effects" the first 2 and last 2 cells never change state.



Figure 2

Continued on Page 26

Continued from Page 25

Figure 2 shows a printer dump (not available in the Basic program) of the totalistic automaton

20 = 010100

for a random set of 160 cells in the first generation. Out of the random initial state we see 2 almost successful growths of new Aliens, and one rather awesome Alien with the disturbing property of reproducing itself every 23 generations. This behavior lies somewhere in the conceptual region between complete chaos on one hand, and trivial order on the other. It is believed that this automaton is the simplest structure having this subjective property, which is causing some excitement even outside the computer community.



Figure 3

Figure 3 shows another automaton which can be calculated with the Basic program, using code

41 = 101001.

Automata such as this have the property that if the parent cells are all empty, then the daughter cell always contains a spore. This is not a very good simulation of real life but often leads to amusing patterns. Explore, explore! Figure 4 shows a Space Invader grown with automaton

0000123002130012300000

using 7 parent cells and 3 kinds of spores.



Figure 4

Can such automata tell us something about the origin of life? Could life, rather than Alien death, be transmitted through space via spores with such primitive genetic codes? These questions are not the usual stuff of computer games, but your "games" computer may help answer them. Perhaps at least you have caught the "bug" of cellular automata!

Looking at cellular automata from another perspective, we face the down-to-earth question: Is there a growth on the wallpaper?

Continued on Page 27



The Pokey Press

Continued from Page 26

```
10 REM AUTOM.BAS
20 REM BY KIRK MCDONALD J.A.C.G.
30 DIM A(19),B$(99),C(159),D(159),F(5)
40 DIM R$(1),CC$(1),AA$(1),E$(1),W$(1)
50 R$="R":CC$="C":AA$="A":E$="E"
60 GOTO 380
100 FOR I=0 TO 159:C(I)=D(I):NEXT I
110 ITOT=C(0)+C(1)+C(2)+C(3)+C(4)
120 D(2)=F(ITOT)
130 FOR I=3 TO 157:ITOT=ITOT-C(I-3)+C(I+2):
    D(I)=F(ITOT):NEXT I
140 J=-1
150 FOR IC=0 TO 19:TEMP=0:J=J+8:RESTORE
160 FOR K=0 TO 7:I=J-K:READ SEMP
170 IF D(I)=1 THEN TEMP=TEMP+SEMP
180 NEXT K
190 POKE JPT+IC,TEMP
200 NEXT IC
210 RETURN
220 DATA 1,2,4,8,16,32,64,128
300 J=-1
310 FOR IC=0 TO 19:TEMP=A(IC):J=J+8
320 FOR K=0 TO 7:I=J-K:SEMP=INT(TEMP/2):
    D(I)=TEMP-2*SEMP:TEMP=SEMP:NEXT K
330 NEXT IC
340 RETURN
380 GRAPHICS 6:SETCOLOR 4,0,14:SETCOLOR 0,0,0
390 SCR=PEEK(88)+256*PEEK(89)
400 ICON=0:IR=79
410 ON ICON GOTO 550,540
420 ? "CHOOSE AUTOMATON # FROM 0 TO 63 ";
430 TRAP 460
440 INPUT AUTO
450 IF AUTO>-1 AND AUTO<64 THEN 470
460 ? "TRY AGAIN":GOTO 420
470 TEMP=INT(AUTO):AUTO=TEMP
480 FOR I=0 TO 5
490 SEMP=INT(TEMP/2)
500 F(I)=TEMP-2*SEMP
510 TEMP=SEMP
520 NEXT I
530 FOR IC=0 TO 19:A(IC)=INT(255*RND(DUM)):
    NEXT IC
540 FOR IC=0 TO 19:POKE SCR+IC,A(IC):NEXT IC
550 ? "AUTOMATON = ";AUTO;" (";F(5);F(4);F(3);
    F(2);F(1);F(0);")"
560 IF ICON<>1 THEN GOSUB 300
570 POKE 764,255
580 TRAP 700
590 ? "PRESS RETURN ONCE TO GO TO MENU"
600 IF ICON=1 AND IR<79 THEN 640
610 JPT=SCR:IRMIN=1
620 IF ICON=1 THEN IRMIN=0
630 FOR IR=IRMIN TO 79
640 IF IR>0 THEN JPT=JPT+20
650 GOSUB 100
660 KEY=PEEK(764)
670 IF KEY<255 THEN INPUT B$:GOTO 700
680 NEXT IR
700 ? "RERUN=R, CONTINUE=C,"
710 ? "NEW AUTOMATON=A, END=E ";
740 TRAP 800
750 INPUT W$
760 IF W$=R$ THEN ICON=2:GOTO 840
770 IF W$=CC$ THEN ICON=1:GOTO 830
780 IF W$=AA$ THEN ICON=0:GOTO 840
790 IF W$=E$ THEN 900
800 ? "TRY AGAIN":GOTO 700
830 IF IR<79 THEN 410
840 FOR I=0 TO 1599:POKE SCR+I,0:NEXT I
850 GOTO 410
900 GRAPHICS 0
910 END
```

TRADING POST

Trading Post is a service for JACG members who wish to sell or swap items of any type. There is no charge for this service. Material must reach the Editor by the 20th of the month to be considered for inclusion in the following month's Trading Post. No commercial services or items will be accepted.

>>>>>>>>><<<<<<<<<

FOR SALE: Zenith Data Systems green video monitor, 12". Great for word processing. $45.00 Contact Frank Pazel at (201) 627-8845.

FOR SALE: Atari 850 Interface Module and 830 Acoustic Modem with original downloader program. Asking $190.00 for the pair. Contact Dave Buschke at (201) 366-9730.

JACG MEMBERSHIP

The Jersey Atari Computer Group (JACG) invites you to become a member. Dues are $20.00 per year and entitle the member to: 1) Receive the monthly newsletter; 2) Purchase programs from the group's extensive tape and disk libraries at special rates; 3) Join special interest groups or form new ones; 4) Benefit from the expertise and experience of other Atari computer users; 5) Participate in group purchases of software at substantially reduced prices; 6) Receive a membership card that entitles the member to discounts at local computer stores; 7) Attend monthly meetings to learn about the latest hardware and software, rumors, and techniques for getting the most out of your Atari computer; 8) Submit articles and programs to the newsletter and give demos and presentations at the monthly meetings; 9) Participate in sale/swap activities with other members; 10) Access the JACG nationally famous Bulletin Board; and 11) Have a lot of fun.

If all of this sounds good to you send a check or money order, payable to JACG, to:

Ron Kordos
201 Lake Valley Road
Morristown, NJ 07960

Remember, receiving the JACG newsletter is just one of the many benefits of being a member of JACG.



DON'T FORGET!
Contribute an article this month.